

## **Appendix 4. SAS code for randomly interventional analogues of path-specific effects**

This SAS code is developed to estimate the randomly interventional analogues of path-specific effects (PSE) of exposure A on a continuous outcome Y with continuous mediators  $M_1$  and  $M_2$  under the regression models in the “A regression based approach and illustration” section. Suppose we have a dataset named mydata with outcome variable “Y”, exposure variables “A”, two mediators “M1” and “M2”, and three covariates “c1”, “c2”, and “c3”. If there were more or fewer covariates the user would have to modify the second to seven, thirteen, and fourteen lines of the code below to include these covariates. In the fourth line of code, the user has to specify the two levels of A ( $a_1=1$  and  $a_0=0$ ) that are being compared and the value of the covariates C at which the effects are to be calculated ( $cc_1=1$ ;  $cc_2=50$ ;  $cc_3=0$ ), according to the values in the application of interest. The output demonstrates estimates, confidence intervals, and p-value for the total effect, four rPSEs, the effect via the first mediator (with/without the second one), the effect via the second mediator (with/without the first one), the total indirect effect, and the proportion divided by total effect.

```

proc nlmixed data=mydata; /* specify dataset named "mydata" */

parms t0=0 t1=0 t2=0 t3=0 t4=0 t5=0 tc1=0 tc2=0 tc3=0 b0=0 b1=0 b2=0 b3=0

bc1=0 bc2=0 bc3=0 r0=0 r1=0 rc1=0 rc2=0 rc3=0 ss_m1=1 ss_m2=1 ss_y=1; /*
parameter to be estimated*/

a1=1; a0=0; cc1=1; cc2=50; cc3=0; /*parameter to be intervened*/

/* regression model for mean of all variables */

mu_y=t0 + t1*A + t2*M1 + t3*M2 + t4*A*M1 + t5*A*M2 + tc1*C1 + tc2*C2 +
tc3*C3;

mu_m2 =b0 + b1*A + b2*M1 + b3*A*M1 + bc1*C1 + bc2*C2 + bc3*C3;

mu_m1 =r0 + r1*A + rc1*C1 + rc2*C2 + rc3*C3;

/* score function for all variables*/

ll_y= -((y-mu_y)**2)/(2*ss_y)-0.5*log(ss_y);

ll_m2= -((m2-mu_m2)**2)/(2*ss_m2)-0.5*log(ss_m2);

ll_m1= -((m1-mu_m1)**2)/(2*ss_m1)-0.5*log(ss_m1);

ll_o= ll_m1 + ll_m2 + ll_y;

model Y ~general(ll_o); /* estimate parameters */

```

```

/* calculate all estimate we want */

bcc = bc1*cc1 + bc2*cc2 + bc3*cc3;

rcc = rc1*cc1 + rc2*cc2 + rc3*cc3;

pse0 =

((t1+t5*(b0+b2*r0+b2*rcc+bcc)+t4*(r0+rcc))+(t4*r1+t5*(b1+b3*r0+b3*rcc
+b2*r1))*a0+t5*b3*r1*a0*a0)*(a1-a0);

pse1=(t2*r1+t4*r1*a1)*(a1-a0);

pse2

=(t3*(b1+b3*r0+b3*rcc)+t5*(b1+b3*r0+b3*rcc)*a1+t3*b3*r1*a0+t5*b3*r1*a
1*a0)*(a1-a0);

pse12 =(t3*b2*r1+((t5*b2*r1+t3*b3*r1))*a1+t5*b3*r1*a1*a1)*(a1-a0);

ie1=pse1+pse12;

ie2=pse2+pse12;

ie = pse1+pse2+pse12;

te = pse0+ie;

estimate 'Direct Effect' pse0;

estimate 'Path Specific Effect via M1 alone' pse1;

estimate 'Path Specific Effect via M2 alone' pse2;

```

```
estimate 'Path Specific Effect via both M1 and M2' pse12;

estimate 'Path Specific Effect via M1 (with/out M2)' ie1;

estimate 'Path Specific Effect via M2 (with/out M1)' ie2;

estimate 'Total Indirect Effect' ie;

estimate 'Total Effect' te;

estimate 'Proportion Direct Effect' pse0/te;

estimate 'Proportion via M1' pse1/te;

estimate 'Proportion via M2' pse2/te;

estimate 'Proportion via both M1 and M2' pse12/te;

estimate 'Total Proportion via M1' ie1/te;

estimate 'Total Proportion via M2' ie2/te;

estimate 'Proportion via M1 or M2' ie/te;

run;
```